

Laboratorio de Algoritmos y Patrones: Neo4j

Departamento de Informática

22 de abril de 2026

1 Introducción al Laboratorio

En este laboratorio exploraremos por qué Neo4j es la base de datos preferida para sistemas de recomendación y análisis de redes. Aprenderemos a pensar en **patrones** y **recorridos** en lugar de tablas.

1.1 Patrón 1: Recomendación por «Sabiduría de la Red»

Escenario: Queremos recomendar a Víctor guías que sus amigos directos han valorado muy positivamente (5 estrellas), pero que él todavía no conoce.

Consulta de Recomendación

```
MATCH (yo:Usuario {nombre: 'Víctor'})-[:SIGUE]->(amigo)-[v:VALORO]->(g:Guia)
WHERE v.estrellas = 5
      AND NOT (yo)-[:VALORO]->(g)
RETURN g.nombre AS Guia_Sugerido,
       amigo.nombre AS Recomendado_por,
       g.especialidad AS Especialidad
ORDER BY g.nombre;
```

Análisis Pedagógico: Fíjate cómo el lenguaje Cypher permite expresar la negación (AND NOT) de una relación de forma natural. En SQL, esto requeriría una subconsulta con NOT EXISTS muy ineficiente.

1.2 Patrón 2: Caminos Cortos (Grados de Separación)

Escenario: ¿Cómo de lejos está Víctor de Elena en la red social? ¿Quién es el «puente» entre ellos?

Algoritmo de Camino Más Corto

```
MATCH p=shortestPath(
  (v:Usuario {nombre: 'Víctor'})-[:SIGUE*..10]->(e:Usuario {nombre: 'Elena'})
)
RETURN p;
```

Reflexión: La función `shortestPath` es un algoritmo interno de Neo4j optimizado para recorrer millones de nodos en milisegundos. El parámetro `*..10` indica que busque hasta un máximo de 10 saltos de distancia.

1.3 Patrón 3: Análisis de Influencia (Centralidad Básica)

Escenario: Queremos saber quién es el usuario más influyente (el que tiene más seguidores) y quién es el guía mejor valorado de media.

Agregación en Grafos

```
// Top Influencers
MATCH (u:Usuario)<-[:SIGUE]-(seguidor)
RETURN u.nombre, count(seguidor) AS num_seguidores
ORDER BY num_seguidores DESC;

// Top Guías por puntuación media
MATCH (g:Guia)<-[v:VALORO]-()
RETURN g.nombre, avg(v.estrellas) AS media_estrellas
ORDER BY media_estrellas DESC;
```

1.4 Laboratorio de Depuración: Errores de Grafo

Peligro: El Producto Cartesiano

Si haces un `MATCH` de dos nodos sin una relación que los una, Neo4j intentará emparejar CADA nodo A con CADA nodo B.

```
MATCH (u:Usuario), (g:Guia)
RETURN u, g; // ¡PELIGRO! Si tienes 1000 usuarios y 1000 guías, esto devuelve
1.000.000 de resultados.
```

Solución: Asegúrate siempre de que tus patrones de `MATCH` estén conectados por flechas a menos que busques explícitamente una comparativa global.